4.3 The method of undetermined coefficients

See table 0.3 on page 12 of Powers for the form of **paticular solution**

```
> restart;
```

Example1:

Consider the equation

```
> eq:= diff(y(t),t$4) - 4*diff(y(t),t$2) = t^2 + exp(t);
```

$$eq := (\frac{\partial^4}{\partial t^4}\,\mathrm{y}(t)) - 4\,(\frac{\partial^2}{\partial t^2}\,\mathrm{y}(t)) = t^2 + e^t$$

Characteristic equation

```
> chareq:= r^4 - 4*r^2 = 0;
```

$$chareq := r^4 - 4\,r^2 = 0$$

Characteristic roots

```
> chr := solve(chareq,r);
```

$$chr := 0,\, 0,\, 2,\, -2$$

Fundamental solutions for the homogeneous equation

For r = 0: s = 2 because 0 appears twice in the above list. We have two solutions corresponding to this root (Note : exp(0*t) = 1)

```
> y1:= t->1; y2:= t->t;
```

$$y1 := 1$$

$$y2 := t \to t$$

For r = 2 or r = -2: single roots so that s=1. We have two corresponding solutions

```
> y3:= t->exp(2*t); y4:= t->exp(-2*t);
```

$$y3 := t \to e^{(2\,t)}$$

$$y4 := t \to e^{(-2\,t)}$$

Particular solution:

Because the terms on the right hand side of eq are not of the same form so that we can use the table right away. We need to split them into 2 eqns. (in Maple: **lhs(eq)** and **rhs(eq)** are **respectively** the **left and right hand sides** of the variable **eq**.

```
> eq1:= lhs(eq)=t^2; eq2:= lhs(eq)=exp(t);
```

$$eq1 := (\frac{\partial^4}{\partial t^4}\,\mathrm{y}(t)) - 4\,(\frac{\partial^2}{\partial t^2}\,\mathrm{y}(t)) = t^2$$

$$eq2 := (\frac{\partial^4}{\partial t^4}\,\mathrm{y}(t)) - 4\,(\frac{\partial^2}{\partial t^2}\,\mathrm{y}(t)) = e^t$$

For the equation eq1:

The right hand side of **eq1** is a polynomial of degree 2 so that the look up table 4.3.1 tells us that the form of a particular solution could be of the form **t^s*(A*t^2 + B*t +C)** with **s is the number of times 0 is a characteristic root**. In our case, **0 is a root and it appears twice** so that s=2. Therefore a particular solution can be found in the form

```
> Y1:= t->t^2*(A*t^2 + B*t + C);
```

We now substitute Y(t) in to the equation **eq1**

```
> subs(y(t)=Y1(t), eq1);
```

```
> simplify(");
```

$$Y1 := t \to t^2\,(A\,t^2 + B\,t + C)$$

$$(\frac{\partial^4}{\partial t^4}\,t^2\,(A\,t^2 + B\,t + C)) - 4\,(\frac{\partial^2}{\partial t^2}\,t^2\,(A\,t^2 + B\,t + C)) = t^2$$

$$24\,A - 48\,A\,t^2 - 24\,B\,t - 8\,C = t^2$$

Collect the like terms and equal the coefficients we have the following equations to determine A,B,C.

```
> ABC:=solve({-48*A = 1, -24*B = 0, 24*A - 8*C =0});
```

Substitute this into Y1:

```
> sol1 := subs(ABC,Y1(t));
```

$$ABC := \{B = 0,\ A = \frac{-1}{48},\ C = \frac{-1}{16}\}$$

$$sol1 := t^2\,(-\frac{1}{48}\,t^2 - \frac{1}{16})$$

<u>For the equation eq2:</u>

The right hand side of **eq2** is a product of a polynomial of degree 0 and exponential so that the look up table 4.3.1 tells us that the form of a particular solution could be of the form **t^s*A*exp(t)** with **s is the number of times alpha=1 is a characteristic root**. In our case, **1 <u>is not a root</u>** so that s=0. Therefore a particular solution can be found in the form

```
> Y2:= t->A*exp(t);
```

We now substitute Y(t) in to the equation **eq1**

```
> subs(y(t)=Y2(t), eq2);
> simplify(");
```

$$Y2 := t \to A\,e^t$$

$$(\frac{\partial^4}{\partial t^4}\,A\,e^t) - 4\,(\frac{\partial^2}{\partial t^2}\,A\,e^t) = e^t$$

$$-3\,A\,e^t = e^t$$

Equal the coefficients we see that A = -1/3.

Substitute this into Y2:

```
> sol2 := subs(A=-1/3,Y2(t));
```

$$sol2 := -\frac{1}{3}\,e^t$$

Finally a particular solution of **eq** is

```
> sol := sol1 + sol2;
```

$$sol := t^2\,(-\frac{1}{48}\,t^2 - \frac{1}{16}) - \frac{1}{3}\,e^t$$

The general solution of **eq** is

```
> gensol := C1*y1(t) + C2*y2(t) + C3*y3(t) + C4*y4(t) + sol;
```

$$gensol := C1 + C2\,t + C3\,e^{(2\,t)} + C4\,e^{(-2\,t)} + t^2\,(-\frac{1}{48}\,t^2 - \frac{1}{16}) - \frac{1}{3}\,e^t$$